Pensieve header: The naive Kh Program - a variant of the program I first wrote in Kyoto in September 2001.

We load a knot theory package only for pre-loaded PD data and for comparisons with known KH results:

```
Once[<< KnotTheory`]
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
PD[Knot[10, 132]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

```
PD[X[4, 2, 5, 1], X[8, 4, 9, 3], X[5, 12, 6, 13], X[15, 18, 16, 19], X[9, 16, 10, 17],
 X[17, 10, 18, 11], X[13, 20, 14, 1], X[19, 14, 20, 15], X[11, 6, 12, 7], X[2, 8, 3, 7]]
```

```
Alexander[Knot[10, 132]][t]
```

$$1+\frac{1}{t^2}-\frac{1}{t}-t+t^2$$

```
pd = PD[Knot[4, 1]]
```

```
PD[X[4, 2, 5, 1], X[8, 6, 1, 5], X[6, 3, 7, 4], X[2, 7, 3, 8]]
```

```
DrawPD[pd]
```

KnotTheory: DrawPD was written by Emily Redelmeier at the University of Toronto in the summers of 2003 and 2004.



```
Count[{1, 2, 3, 3, 2, 4, 3, 2, 1}, 2]
```

```
3
```

```
n₊[pd_PD] := Count[pd, X[i_, j_, k_, l_] /; j - l == 1 ∨ l - j > 1];
n₋[pd_PD] := Count[pd, X[i_, j_, k_, l_] /; l - j == 1 ∨ j - l > 1];
```

```
{n₊[pd], n₋[pd]}
```

```
{2, 2}
```

```
SetAttributes[p, Orderless]
```

```
{0, 1, 1, 0, 1}
```

```
{0, 1, 1, 0, 1}
```

```
"010001"
```

```
010001
```

```
Characters["Yusuke"]
```

```
{Y, u, s, u, k, e}
```

```
S[pd_PD, s_String] := S[pd, Characters[s] /. {"0" → 0, "1" → 1, "*" → ⋆}]
```

```
S[pd, "01*0"]
```

```
S[PD[X[4, 2, 5, 1], X[8, 6, 1, 5], X[6, 3, 7, 4], X[2, 7, 3, 8]], {0, 1, ⋆, 0}]
```

```
m_ ⋄ n_ := Min[m, n];
```

```
a = {0, 1, 0, 0}
```

```
{0, 1, 0, 0}
```

```
{List @@ pd, a} // MatrixForm
```

$$\begin{pmatrix} X[4,2,5,1] & X[8,6,1,5] & X[6,3,7,4] & X[2,7,3,8] \\ 0 & 1 & 0 & 0 \end{pmatrix}$$

```
{List @@ pd, a}^T
```

```
{{X[4, 2, 5, 1], 0}, {X[8, 6, 1, 5], 1}, {X[6, 3, 7, 4], 0}, {X[2, 7, 3, 8], 0}}
```

```
{List @@ pd, a}^T
```

```
Times @@ ({List @@ pd, a}^T /. {
    {X[i_, j_, k_, l_], 0} :> p[i, j]_{i⋄j} p[k, l]_{k⋄l},
    {X[i_, j_, k_, l_], 1} :> p[i, l]_{i⋄l} p[j, k]_{j⋄k},
    {x_X, ⋆} :> x}
 )
```

$$p[1,5]_1\, p[1,6]_1\, p[2,4]_2\, p[2,7]_2\, p[3,6]_3\, p[3,8]_3\, p[4,7]_4\, p[5,8]_5$$

```
Times @@ ({List @@ pd, a}^T /. {
     {X[i_, j_, k_, l_], 0} :> p[i, j]_{i⋄j} p[k, l]_{k⋄l},
     {X[i_, j_, k_, l_], 1} :> p[i, l]_{i⋄l} p[j, k]_{j⋄k},
     {x_X, ⋆} :> x}
  ) //. {
  p[i_, j_]_m_ p[j_, k_]_n_ :> p[i, k]_{m⋄n}
 }
```

$$p[4,4]_2\, p[6,6]_1$$

```
a = {0, 1, ⋆, 0};
Times @@ ({List @@ pd, a}ᵀ /. {
      {X[i_, j_, k_, l_], 0} :> p[i, j]_{i◇j} p[k, l]_{k◇l},
      {X[i_, j_, k_, l_], 1} :> p[i, l]_{i◇l} p[j, k]_{j◇k},
      {x_X, ⋆} :> x}
  ) //. {
  p[i_, j_]_m_ p[j_, k_]_n_ :> p[i, k]_{m◇n}
 }
```

```
p[3, 6]_1 p[4, 7]_2 X[6, 3, 7, 4]
```

```
S[pd_PD, a_List] := Times @@ ({List @@ pd, a}ᵀ /. {
         {X[i_, j_, k_, l_], 0} :> p[i, j]_{i◇j} p[k, l]_{k◇l},
         {X[i_, j_, k_, l_], 1} :> p[i, l]_{i◇l} p[j, k]_{j◇k},
         {x_X, ⋆} :> x}
     ) //. {
     p[i_, j_]_m_ p[j_, k_]_n_ :> p[i, k]_{m◇n}
    } //. {
    X[i_, j_, k_, l_] p[i_, j_]_m_ p[k_, l_]_n_ :> (c_m c_n → c_{m◇n}),
    X[i_, j_, k_, l_] p[i_, l_]_m_ p[j_, k_]_n_ :> (c_{m◇n} → c_m c_n)
   } //. p[__]_m_ :> c_m
```

```
S[pd, {1, 0, 0, 0}]
```

```
c_1 c_3
```

```
{S[pd, "0100"], S[pd, "0110"], S[pd, "01⋆0"]}
```

```
{c_1 c_2, c_1, c_1 c_2 → c_1}
```

```
{S[pd, "0000"], S[pd, "1000"], S[pd, "⋆000"]}
```

```
{c_1 c_2 c_3, c_1 c_3, (c_1 c_2 → c_1) c_3}
```

```
{S[pd, "1100"], S[pd, "1110"], S[pd, "11⋆0"]}
```

```
{c_1, c_1 c_2, c_1 → c_1 c_2}
```

```
a = {0, 1, 0, 0}
```

```
{0, 1, 0, 0}
```

```
S[pd, a]
```

```
c_1 c_2
```

```
S[pd, a] /. c_x_ :> (vp_x + vm_x)
```

```
(vm_1 + vp_1) (vm_2 + vp_2)
```

```
Expand[S[pd, a] /. c_x_ :> (vp_x + vm_x)]
```

$vm_1\ vm_2 + vm_2\ vp_1 + vm_1\ vp_2 + vp_1\ vp_2$

```
V[pd_PD, a_] := List @@ Expand[S[pd, a] /. c_x_ :> (vp_x + vm_x)]
```

```
V[pd, "0100"]
```

$\{vm_1\ vm_2,\ vm_2\ vp_1,\ vm_1\ vp_2,\ vp_1\ vp_2\}$

```
d[pd_PD, a_] := S[pd, a] /. {
    (c_x_ c_y_ -> c_z_) * _. :> {vp_x vp_y -> vp_z, vp_x vm_y -> vm_z, vm_x vp_y -> vm_z, vm_x vm_y -> 0},
    (c_z_ -> c_x_ c_y_) * _. :> {vp_z -> vp_x vm_y + vm_x vp_y, vm_z -> vm_x vm_y}
  }
```

```
{S[pd, "01*0"], d[pd, "01*0"]}
```

$\{c_1\ c_2 \to c_1,\ \{vp_1\ vp_2 \to vp_1,\ vm_2\ vp_1 \to vm_1,\ vm_1\ vp_2 \to vm_1,\ vm_1\ vm_2 \to 0\}\}$

```
{S[pd, "*000"], d[pd, "*000"]}
```

$\{(c_1\ c_2 \to c_1)\ c_3,\ \{vp_1\ vp_2 \to vp_1,\ vm_2\ vp_1 \to vm_1,\ vm_1\ vp_2 \to vm_1,\ vm_1\ vm_2 \to 0\}\}$

```
{S[pd, "11*0"], d[pd, "11*0"]}
```

$\{c_1 \to c_1\ c_2,\ \{vp_1 \to vm_2\ vp_1 + vm_1\ vp_2,\ vm_1 \to vm_1\ vm_2\}\}$

```
{S[pd, "*000"] , V[pd, "0000"], d[pd, "*000"]} // Column
```

$(c_1\ c_2 \to c_1)\ c_3$
$\{vm_1\ vm_2\ vm_3,\ vm_2\ vm_3\ vp_1,\ vm_1\ vm_3\ vp_2,\ vm_3\ vp_1\ vp_2,\ vm_1\ vm_2\ vp_3,\ vm_2\ vp_1\ vp_3,\ vm_1\ vp_2\ vp_3,\ vp_1\ vp_2\ vp_3\}$
$\{vp_1\ vp_2 \to vp_1,\ vm_2\ vp_1 \to vm_1,\ vm_1\ vp_2 \to vm_1,\ vm_1\ vm_2 \to 0\}$

```
{V[pd, "0000"], V[pd, "0000"] /. d[pd, "*000"]} // MatrixForm
```

$$\begin{pmatrix} vm_1\ vm_2\ vm_3 & vm_2\ vm_3\ vp_1 & vm_1\ vm_3\ vp_2 & vm_3\ vp_1\ vp_2 & vm_1\ vm_2\ vp_3 & vm_2\ vp_1\ vp_3 & vm_1\ vp_2\ vp_3 & vp_1\ vp_2\ vp_3 \\ 0 & vm_1\ vm_3 & vm_1\ vm_3 & vm_3\ vp_1 & 0 & vm_1\ vp_3 & vm_1\ vp_3 & vp_1\ vp_3 \end{pmatrix}$$

```
Exponent[q^7, q]
```

7

```
udeg[P_] := Exponent[P /. {v_a_ :> q^Total[a], vp_ -> q, vm_ -> q^-1}, q]
```

```
udeg[vm1 vm2 vp3]
```

-1

```
udeg[v_{0,0,0,1,1} vm1]
```

1

```
Table[0, 3]
```

$\{0, 0, 0\}$

```
Table[1, 2]
```

```
{1, 1}
```

```
Table[0, 3] ~Join~ Table[1, 2]
```

```
{0, 0, 0, 1, 1}
```

```
Permutations[{1, 2, 3, 4}]
```

```
{{1, 2, 3, 4}, {1, 2, 4, 3}, {1, 3, 2, 4}, {1, 3, 4, 2}, {1, 4, 2, 3}, {1, 4, 3, 2},
 {2, 1, 3, 4}, {2, 1, 4, 3}, {2, 3, 1, 4}, {2, 3, 4, 1}, {2, 4, 1, 3}, {2, 4, 3, 1},
 {3, 1, 2, 4}, {3, 1, 4, 2}, {3, 2, 1, 4}, {3, 2, 4, 1}, {3, 4, 1, 2}, {3, 4, 2, 1},
 {4, 1, 2, 3}, {4, 1, 3, 2}, {4, 2, 1, 3}, {4, 2, 3, 1}, {4, 3, 1, 2}, {4, 3, 2, 1}}
```

```
Permutations[{1, 2, 3, 4}] // Length
```

```
24
```

```
Permutations[Table[i, {i, 11}]] // Length
```

```
39 916 800
```

```
11!
```

```
39 916 800
```

```
Permutations[Table[0, 3] ~Join~ Table[1, 2]]
```

```
{{0, 0, 0, 1, 1}, {0, 0, 1, 0, 1}, {0, 0, 1, 1, 0}, {0, 1, 0, 0, 1}, {0, 1, 0, 1, 0},
 {0, 1, 1, 0, 0}, {1, 0, 0, 0, 1}, {1, 0, 0, 1, 0}, {1, 0, 1, 0, 0}, {1, 1, 0, 0, 0}}
```

```
With[{r = 1},
 Permutations[Table[0, n+[pd] - r] ~Join~ Table[1, r + n-[pd]]]
]
```

```
{{0, 1, 1, 1}, {1, 0, 1, 1}, {1, 1, 0, 1}, {1, 1, 1, 0}}
```

```
f[x_] := x^2
```

```
f[3]
```

```
9
```

```
f /@ {1, 2, 3, 4, 5}
```

```
{1, 4, 9, 16, 25}
```

```
(#^2 + #) & /@ {1, 2, 3, 4, 5}
```

```
{2, 6, 12, 20, 30}
```

```
With[{r = 1},
 (((v_#) V[pd, #]) & /@ Permutations[Table[0, n+[pd] - r] ~Join~ Table[1, r + n-[pd]]])
]
```

$$\{\{v_{\{0,1,1,1\}}\, vm_1\, vm_3,\ v_{\{0,1,1,1\}}\, vm_3\, vp_1,\ v_{\{0,1,1,1\}}\, vm_1\, vp_3,\ v_{\{0,1,1,1\}}\, vp_1\, vp_3\},$$
$$\{v_{\{1,0,1,1\}}\, vm_1\, vm_3,\ v_{\{1,0,1,1\}}\, vm_3\, vp_1,\ v_{\{1,0,1,1\}}\, vm_1\, vp_3,\ v_{\{1,0,1,1\}}\, vp_1\, vp_3\},$$
$$\{v_{\{1,1,0,1\}}\, vm_1\, vm_2,\ v_{\{1,1,0,1\}}\, vm_2\, vp_1,\ v_{\{1,1,0,1\}}\, vm_1\, vp_2,\ v_{\{1,1,0,1\}}\, vp_1\, vp_2\},$$
$$\{v_{\{1,1,1,0\}}\, vm_1\, vm_2,\ v_{\{1,1,1,0\}}\, vm_2\, vp_1,\ v_{\{1,1,1,0\}}\, vm_1\, vp_2,\ v_{\{1,1,1,0\}}\, vp_1\, vp_2\}\}$$

```
KC[pd_PD, r_] := If[r < -n-[pd] || r > n+[pd], {},
    Join @@
     (((v_#) V[pd, #]) & /@ Permutations[Table[0, n+[pd] - r] ~Join~ Table[1, r + n-[pd]]])
   ];
KC[pd_PD, r_, deg_] := Cases[KC[pd, r], u_ /; udeg[u] - 2 n-[pd] + n+[pd] == deg]
```

```
KC[pd, 1]
```

$$\{v_{\{0,1,1,1\}}\, vm_1\, vm_3,\ v_{\{0,1,1,1\}}\, vm_3\, vp_1,\ v_{\{0,1,1,1\}}\, vm_1\, vp_3,\ v_{\{0,1,1,1\}}\, vp_1\, vp_3,$$
$$v_{\{1,0,1,1\}}\, vm_1\, vm_3,\ v_{\{1,0,1,1\}}\, vm_3\, vp_1,\ v_{\{1,0,1,1\}}\, vm_1\, vp_3,\ v_{\{1,0,1,1\}}\, vp_1\, vp_3,$$
$$v_{\{1,1,0,1\}}\, vm_1\, vm_2,\ v_{\{1,1,0,1\}}\, vm_2\, vp_1,\ v_{\{1,1,0,1\}}\, vm_1\, vp_2,\ v_{\{1,1,0,1\}}\, vp_1\, vp_2,$$
$$v_{\{1,1,1,0\}}\, vm_1\, vm_2,\ v_{\{1,1,1,0\}}\, vm_2\, vp_1,\ v_{\{1,1,1,0\}}\, vm_1\, vp_2,\ v_{\{1,1,1,0\}}\, vp_1\, vp_2\}$$

```
KC[pd, 1, 1]
```

$$\{v_{\{0,1,1,1\}}\, vm_3\, vp_1,\ v_{\{0,1,1,1\}}\, vm_1\, vp_3,\ v_{\{1,0,1,1\}}\, vm_3\, vp_1,\ v_{\{1,0,1,1\}}\, vm_1\, vp_3,$$
$$v_{\{1,1,0,1\}}\, vm_2\, vp_1,\ v_{\{1,1,0,1\}}\, vm_1\, vp_2,\ v_{\{1,1,1,0\}}\, vm_2\, vp_1,\ v_{\{1,1,1,0\}}\, vm_1\, vp_2\}$$

```
dd[pd_PD][expr_] := Expand[expr] /. s_ * v_a_ :> Expand[σ = 1;
    Sum[
     If[a[[i]] == 0,
      σ * v_ReplacePart[a,1,i] * s /. d[pd, List @@ ReplacePart[a, ★, i]],
      σ *= -1; 0
     ],
     {i, Length[a]}
    ]
   ]
```

```
KC[pd, 0, 1]
```

$$\{v_{\{0,0,1,1\}}\, vm_3\, vp_1\, vp_2,\ v_{\{0,0,1,1\}}\, vm_2\, vp_1\, vp_3,\ v_{\{0,0,1,1\}}\, vm_1\, vp_2\, vp_3,$$
$$v_{\{0,1,0,1\}}\, vp_1,\ v_{\{0,1,1,0\}}\, vp_1,\ v_{\{1,0,0,1\}}\, vp_1,\ v_{\{1,0,1,0\}}\, vp_1,\ v_{\{1,1,0,0\}}\, vp_1\}$$

```
KC[pd, 0, 1] // Length
```

8

```
KC[pd, 0, 1] // dd[pd]
```

$$\{v_{\{0,1,1,1\}}\, vm_3\, vp_1 + v_{\{1,0,1,1\}}\, vm_3\, vp_1,$$
$$v_{\{0,1,1,1\}}\, vm_1\, vp_3 + v_{\{1,0,1,1\}}\, vm_1\, vp_3,\ v_{\{0,1,1,1\}}\, vm_1\, vp_3 + v_{\{1,0,1,1\}}\, vm_1\, vp_3,$$
$$v_{\{1,1,0,1\}}\, vm_2\, vp_1 - v_{\{0,1,1,1\}}\, vm_3\, vp_1 + v_{\{1,1,0,1\}}\, vm_1\, vp_2 - v_{\{0,1,1,1\}}\, vm_1\, vp_3,$$
$$v_{\{1,1,1,0\}}\, vm_2\, vp_1 + v_{\{0,1,1,1\}}\, vm_3\, vp_1 + v_{\{1,1,1,0\}}\, vm_1\, vp_2 + v_{\{0,1,1,1\}}\, vm_1\, vp_3,$$
$$-v_{\{1,1,0,1\}}\, vm_2\, vp_1 - v_{\{1,0,1,1\}}\, vm_3\, vp_1 - v_{\{1,1,0,1\}}\, vm_1\, vp_2 - v_{\{1,0,1,1\}}\, vm_1\, vp_3,$$
$$-v_{\{1,1,1,0\}}\, vm_2\, vp_1 + v_{\{1,0,1,1\}}\, vm_3\, vp_1 - v_{\{1,1,1,0\}}\, vm_1\, vp_2 + v_{\{1,0,1,1\}}\, vm_1\, vp_3,$$
$$v_{\{1,1,0,1\}}\, vm_2\, vp_1 + v_{\{1,1,1,0\}}\, vm_2\, vp_1 + v_{\{1,1,0,1\}}\, vm_1\, vp_2 + v_{\{1,1,1,0\}}\, vm_1\, vp_2\}$$

```
KC[pd, 0, 1] // dd[pd] // dd[pd]
```

$$\{0, 0, 0, 0, 0, 0, 0, 0\}$$

```
Rank[pd_PD, r_, deg_] := (
   B0 = KC[pd, r, deg];
   B1 = KC[pd, r + 1, deg];
   If[B0 == {} ∨ B1 == {}, 0,
    dB0 = dd[pd][B0];
    MatrixRank[Table[Coefficient[db0, b1], {db0, dB0}, {b1, B1}]]
   ]
  );
```

```
Rank[pd, 1, 1]
```

$$3$$

```
Betti[pd_PD, r_, deg_] := Length[KC[pd, r, deg]] - Rank[pd, r, deg] - Rank[pd, r - 1, deg]
```

```
Betti[pd, 1, 1]
```

$$1$$

```
Kh1[pd_PD] := Sum[
  t^r q^deg Betti[pd, r, deg],
  {r, -n_[pd], n_+[pd]},
  {deg, Union[udeg /@ KC[pd, r]] - 2 n_[pd] + n_+[pd]}
 ]
```

```
Kh1[pd]
```

$$\frac{1}{q} + q + \frac{1}{q^5\, t^2} + \frac{1}{q\, t} + q\, t + q^5\, t^2$$

```
Kh1[PD[Knot[8, 17]]]
```

$$\frac{4}{q} + 4\, q + \frac{1}{q^9\, t^4} + \frac{2}{q^7\, t^3} + \frac{1}{q^5\, t^3} + \frac{3}{q^5\, t^2} + \frac{2}{q^3\, t^2} +$$
$$\frac{3}{q^3\, t} + \frac{3}{q\, t} + 3\, q\, t + 3\, q^3\, t + 2\, q^3\, t^2 + 3\, q^5\, t^2 + q^5\, t^3 + 2\, q^7\, t^3 + q^9\, t^4$$

```
Timing@Table[
  K → Kh[K][q, t] == Kh1[PD@K],
  {K, AllKnots[{3, 6}]}
 ]
```

KnotTheory: Loading precomputed data in Kh4Knots`.

```
{6.39063, {Knot[3, 1] → True, Knot[4, 1] → True, Knot[5, 1] → True,
  Knot[5, 2] → True, Knot[6, 1] → True, Knot[6, 2] → True, Knot[6, 3] → True}}
```

```
m_ ⋄ n_ := Min[m, n];
Kh2[K_] := Module[{pd, np, nm, p, S, a, ⋆, c, V,
    vp, vm, d, udeg, KC, v, dd, σ, Rank, B0, B1, dB0, db0, b1, Betti},
   pd = PD[K];
   np = Count[pd, X[i_, j_, k_, l_] /; j - l == 1 ∨ l - j > 1];
   nm = Count[pd, X[i_, j_, k_, l_] /; l - j == 1 ∨ j - l > 1];
   SetAttributes[p, Orderless];
   S[a_List] := S[a] = Times @@ ({List @@ pd, a}ᵀ /. {
             {X[i_, j_, k_, l_], 0} :> p[i, j]_{i⋄j} p[k, l]_{k⋄l},
             {X[i_, j_, k_, l_], 1} :> p[i, l]_{i⋄l} p[j, k]_{j⋄k},
             {x_X, ⋆} :> x}
         ) //. {
         p[i_, j_]_m_ p[j_, k_]_n_ :> p[i, k]_{m⋄n}
        } //. {
        X[i_, j_, k_, l_] p[i_, j_]_m_ p[k_, l_]_n_ :> (c_m c_n → c_{m⋄n}),
        X[i_, j_, k_, l_] p[i_, l_]_m_ p[j_, k_]_n_ :> (c_{m⋄n} → c_m c_n)
       } //. p[__]_m_ :> c_m;
   V[a_] := V[a] = List @@ Expand[S[a] /. c_x_ :> (vp_x + vm_x)];
   d[a_] := d[a] = S[a] /. {
        (c_x_ c_y_ → c_z_) * _. :> {vp_x vp_y → vp_z, vp_x vm_y → vm_z, vm_x vp_y → vm_z, vm_x vm_y → 0},
        (c_z_ → c_x_ c_y_) * _. :> {vp_z → vp_x vm_y + vm_x vp_y, vm_z → vm_x vm_y}
       };
   udeg[P_] := Exponent[P /. {v_a_ :> q^Total[a], vp_ → q, vm_ → q^-1}, q];
   KC[r_] := KC[r] = If[r < -nm || r > np, {},
      Join @@ (((v_#) V[#]) & /@ Permutations[Table[0, np - r] ~Join~ Table[1, r + nm]])
     ];
   KC[r_, deg_] := KC[r, deg] = Cases[KC[r], u_ /; udeg[u] - 2 nm + np == deg];
   dd[expr_] := Expand[expr] /. s_ * v_a_ :> Expand[σ = 1;
       Sum[
        If[a[[i]] == 0, σ * v_{ReplacePart[a,1,i]} * s /. d[List @@ ReplacePart[a, ⋆, i]], σ *= -1;
         0], {i, Length[a]}
       ]
      ];
   Rank[r_, deg_] := Rank[r, deg] = (
      B0 = KC[r, deg];
```

```
      B1 = KC[r + 1, deg];
      If[B0 == {} ∨ B1 == {}, 0,
       dB0 = dd[B0];
       MatrixRank[Table[Coefficient[db0, b1], {db0, dB0}, {b1, B1}]]
      ]
     );
   Betti[r_, deg_] := Length[KC[r, deg]] - Rank[r, deg] - Rank[r - 1, deg];
   Sum[
    t^r q^deg Betti[r, deg],
    {r, -nm, np},
    {deg, Union[udeg /@ KC[r]] - 2 nm + np}
   ]
  ]
```

```
Kh2[Knot[4, 1]]
```

$$\frac{1}{q}+q+\frac{1}{q^5\,t^2}+\frac{1}{q\,t}+q\,t+q^5\,t^2$$

```
Kh2[Knot[8, 17]]
```

$$\frac{4}{q}+4\,q+\frac{1}{q^9\,t^4}+\frac{2}{q^7\,t^3}+\frac{1}{q^5\,t^3}+\frac{3}{q^5\,t^2}+\frac{2}{q^3\,t^2}+$$
$$\frac{3}{q^3\,t}+\frac{3}{q\,t}+3\,q\,t+3\,q^3\,t+2\,q^3\,t^2+3\,q^5\,t^2+q^5\,t^3+2\,q^7\,t^3+q^9\,t^4$$

```
Timing@Table[
  K → Kh[K][q, t] == Kh2[K],
  {K, AllKnots[{3, 6}]}
 ]
```

```
{0.84375, {Knot[3, 1] → True, Knot[4, 1] → True, Knot[5, 1] → True,
  Knot[5, 2] → True, Knot[6, 1] → True, Knot[6, 2] → True, Knot[6, 3] → True}}
```

```
{DrawPD[Knot[5, 1]], DrawPD[Knot[10, 132]]}
```



```
Timing[kh12 = {Kh2[Knot[5, 1]], Kh2[Knot[10, 132]]}]
```

$$\left\{103.781,\ \left\{\frac{1}{q^5}+\frac{1}{q^3}+\frac{1}{q^{15}\,t^5}+\frac{1}{q^{11}\,t^4}+\frac{1}{q^{11}\,t^3}+\frac{1}{q^7\,t^2},\right.\right.$$
$$\left.\left.\frac{1}{q^3}+\frac{1}{q}+\frac{1}{q^{15}\,t^7}+\frac{1}{q^{11}\,t^6}+\frac{1}{q^{11}\,t^5}+\frac{1}{q^9\,t^4}+\frac{1}{q^7\,t^4}+\frac{1}{q^9\,t^3}+\frac{1}{q^5\,t^3}+\frac{2}{q^5\,t^2}+\frac{1}{q\,t}\right\}\right\}$$

```
Cancel[kh12 /. t → -1 / (q + q^-1)]
```

$$\left\{\frac{-1+q^2-q^4+q^6+q^{10}}{q^{14}},\ \frac{-1+q^2-q^4+q^6+q^{10}}{q^{14}}\right\}$$

```
{Jones[Knot[5, 1]][q^2], Jones[Knot[10, 132]][q^2]}
```

KnotTheory: Loading precomputed data in Jones4Knots`.

$$\left\{-\frac{1}{q^{14}}+\frac{1}{q^{12}}-\frac{1}{q^{10}}+\frac{1}{q^8}+\frac{1}{q^4},\ -\frac{1}{q^{14}}+\frac{1}{q^{12}}-\frac{1}{q^{10}}+\frac{1}{q^8}+\frac{1}{q^4}\right\}$$